

SCHÜLER EXPERIMENTIEREN
1982

Bereich: Technik

Thema: BAU EINES PLOTTERS aus Fischer-Technik
und Erstellung von Demonstrationspro-
grammen in BASIC für den Betrieb des
Plotters

Inhaltsverzeichnis

Zusammenfassung

1. Einleitung	S. 1
2. Hardware	S. 1
3. Software	S. 5
4. Literatur und Hilfen	S. 9
5. Anhang	S. 9 (A 1 - A 20)

Teilnehmer: Heiko Purnhagen, Bremen, Elsasser Str. 64
(geb. 2.4.69) Schüler der Klasse 7 c am
Gymnasium Hamburger Straße, Bremen

Z U S A M M E N F A S S U N G

In meiner Arbeit beschreibe ich den Aufbau und die Funktionsweise meines Plotters, den ich aus Fischer-Technik gebaut habe, und die dazugehörigen Programme in BASIC, die den Plotter ansteuern.

Der Schreibstift des Plotters wird mit selbstgebauten Schrittmotoren (mit Nockenscheiben) bewegt. Dieses Prinzip funktioniert relativ gut und erlaubt bei einer Schreibgeschwindigkeit von etwa 2 mm/s eine Genauigkeit von 1 mm. Mein Plotter arbeitet auf Papier des Formates DIN A 4. Dieses ergibt ein Raster von 290 Punkten in der X-Achse und 200 Punkten in der Y-Achse. Diese Auflösung genügt, um Geraden, Kreise und andere Figuren vernünftig auszuploten. Als Schreibstift benutze ich einen Filzstift, der von einem Elektromagneten auf das Papier gedrückt werden kann.

Die Ansteuerung des Plotters erfolgt durch BASIC-Programme über ein 8 Bit-Parallelinterface. (Mein Vater besitzt einen Mikrocomputer 8085 mit BASIC.) Man benötigt dafür zwei verschiedene Programmteile, und zwar erstens ein Programm, das die Kurven berechnet. Dieses muß für jeden einzelnen Kurvenpunkt geschehen. Zweitens braucht man ein Programm, das die X-Y-Motoren, die Richtungsrelais und den Schreibstift ansteuert.

Im Laufe meiner Arbeit mit dem Plotter habe ich schon viele verschiedene interessante Figuren und Funktionen gezeichnet.

1. EINLEITUNG

Schon seit längerer Zeit hatte ich Lust dazu, einmal ein computergesteuertes Modell zu bauen (mein Vater besitzt einen Mikrocomputer 8085 mit BASIC). So kam mir die Idee, einen Plotter, d.h. eine numerisch gesteuerte Zeichenmaschine zu konstruieren. Vom Prinzip ist eine solche Zeichenmaschine aus Fischer-Technik relativ einfach zu bauen. So begann ich Ende August 1981 mit den ersten Konstruktionsversuchen. Mein Plotter arbeitet mit zwei selbstgebauten Schrittmotoren (mit Nockenscheiben), die einen Zeichenstift in X- und Y-Richtung über das Papier bewegen. Die Steuerung des Plotters erfolgt über ein BASIC-Programm.

2. HARDWARE

2.1. Aufgabe

Als Konstruktionsmaterial habe ich Fischer-Technik gewählt, da man damit auch komplizierte Modelle stabil und einfach bauen kann. Der Plotter soll auf möglichst einfache Art und Weise von dem Computer angesteuert werden. Unser Computer hat hierfür je ein 8 Bit Parallel-Eingabe- und Ausgabeter, die über BASIC-Befehle (INP/OUT) angesprochen werden können.

2.2. Probleme

2.2.1. Lösungsmöglichkeiten

Als mögliche Konzepte für den X/Y-Antrieb meines Plotters boten sich folgende an:

Entweder baut man den Plotter mit zwei Poti-Regelkreisen auf (s. Aufsatz CHIP 5/81 S. 37/38), oder man benutzt zwei Schrittmotore. Schrittmotore kann man sich entweder kaufen, oder man konstruiert sie selbst aus einem Gleichstrommotor, der mit einer Nockenscheibe geschaltet wird. Ich habe das Prinzip der Schrittmotoren mit Nockenscheiben gewählt, weil das Prinzip mit dem Poti-Regelkreis für mich ein zu hoher Aufwand bedeutet hätte, und da gekaufte Schrittmotoren zu teuer sind und sich in Fischer-Technik schwer einbauen lassen.

2.2.2. Nockenscheibenprinzip

Als ich ein erstes Modell mit Nockenscheiben-Schrittmotorenantrieb fertiggestellt hatte, ergaben sich folgende Probleme:

Entweder drehte sich der Motor, der die Nockenscheibe antrieb, zu schnell (Nockenscheibe überdrehte, dadurch mehr als eine Umdrehung je Startimpuls), oder er drehte sich zu langsam (Anlaufschwierigkeiten des Motors).

Obwohl ich die Motorspannung, die Startimpulsdauer und die Impulspause des Computers einjustieren konnte, war es sehr schwer, für längere Zeit fehlerfrei zu plotten.

Die Übersetzung von der Motorwelle zur Nocke sollte möglichst groß sein, da dann der Motor mehr Zeit zum sicheren Abbremsen hat. Dadurch wurde jedoch bei gegebener Schrittweite die Geschwindigkeit geringer.

2.2.3. Lösungsversuche mit dem Nockenscheibenprinzip

Nachdem ich durch längeres Probieren eine möglichst große Leichtgängigkeit der Mechanik erreicht hatte, funktionierte der Plotter immer noch nicht einwandfrei. Ich habe versucht, dieses Problem mit an verschiedenen Stellen der Nockenscheibe angebrachten Tesa-bandstreifen zu lösen. Dieses brachte aber keinen Erfolg. Auch habe ich versucht, eine Rückmeldung vom Plotter zum Computer zu installieren. Aber dafür war der BASIC-Interpreter zu langsam. Zum Testen habe ich mir aus einem Labor der Hochschule für Technik stabilisierte Netzteile ausgeliehen.* Bei einem Netzteil pro Motor brachte dieses endlich den gewünschten Erfolg. Es stellte sich heraus, daß die Gleichspannung aus den Fischer-Technik-Trafos nicht stabil genug war (stark belastungsabhängig). Zusätzlich mußte ich auch noch die Stromversorgungskabel auswechseln, weil die zuerst eingebauten einen zu hohen Kabelwiderstand hatten (dadurch bis zu 0,5 V Spannungsabfall).

Bei den Tests mit dieser stabilisierten Spannungsversorgung ergab sich, daß der Plotter ab und zu trotzdem nicht richtig funktionierte, da er mit unterschiedlicher Geschwindigkeit arbeitete. Nach längerem Suchen stellte sich heraus, daß die Dauer der Motorstartimpulse (s. Skizze S. 4) je nach Programm und Zufall zwischen 100 ms bis 200 ms schwankte. Dieses rührte vom BASIC-Interpreter her. Deshalb wird jetzt die Verzögerungszeit durch ein Assembler-Unterprogramm erzeugt (s. S. 5 No. 3.2.).

* Die Kosten für die endgültig verwendeten Netzteile hat mir der Senator für Bildung aufgrund einer allgemeinen Zusage beim Wettbewerb JUGEND FORSCHT 1981 erstattet.

Nach einigem Probieren und Justieren der Spannung funktioniert jetzt der Plotter sehr gut. Es treten nur noch sehr selten kleine Störungen auf, die wahrscheinlich von der Temperatur des Motors oder geringfügiger Änderung der Leichtgängigkeit der Mechanik abhängen.

2.3. Beschreibung der Hardware

Mein Plotter arbeitet mit einem Papier des Formats DIN A 4. Seine Schrittgröße beträgt 1 mm und die Schreibgeschwindigkeit etwa 2 mm/s. Dieses ergibt eine Auflösung von 200 Punkten in der Y-Achse und 290 Punkten in der X-Achse. Diese Grenzen werden durch mein Computerprogramm kontrolliert. Zur Sicherung sind noch für jede Richtung Endlagentaster installiert.

2.3.1. Mechanischer Aufbau aus Fischer-Technik

Der Motor für die X-Achse treibt die Nockenscheibe und über ein Getriebe 2 parallel laufende Ketten auf jeder Seite des Papiertisches an. An diesen Ketten ist ein auf Rollen laufender Schlitten befestigt, den man somit über das Papier bewegen kann. An einer Seite des X-Schlittens ist der Y-Motor angebracht, der auch eine Nockenscheibe antreibt. Über ein Getriebe wird von ihm mit einer Kette der Y-Schlitten auf dem X-Schlitten bewegt. Der Y-Schlitten wird mit Zahnstangen stabil geführt. Auf dem Y-Schlitten ist die Halterung für den Stift (PEN) angebracht. Als Stift benutze ich einen Filzstift, der an seinem oberen Ende eine Halterung aus Klebeband zum Hochheben hat. Der Stift wird durch 2 mit Löchern versehenen Fischer-Technik-Flachsteine geführt. Die Klebebandhalterung liegt auf einem Arm aus Statik-Streben auf. Dieser Arm wird von einem Elektromagneten bewegt, der eine an dem Arm angebrachte kleine Eisenplatte anzieht (siehe Foto im Anhang).

2.3.2. Elektronischer Aufbau


Als Ausgang vom Computer stand mir ein Interface mit 8 parallelen Ausgängen zur Verfügung. Mit diesem Interface kann man über Verstärker Motore, Relais oder Ähnliches ansteuern.

2.3.2.1. Start-Stop-Betrieb der Motoren

An jedem Motor sind über ein Getriebe mit der Übersetzung 10 : 1 Nockenscheiben befestigt, die für die Hälfte einer Umdrehung je einen Taster bedienen. Die Taster sind so geschaltet, daß sie den Motor nach jeder Umdrehung (wenn der Taster wieder gedrückt wird) ausschalten.

Die Motoren werden durch Überbrücken des Tasters gestartet. Dieses Überbrücken geschieht mit Hilfe eines Verstärkers, der an einem Ausgang des Interfaces angeschlossen ist. Durch dieses Startsignal, das von begrenzter Dauer ist, beginnt der Motor sich zu drehen. Er stoppt wieder, wenn der Nockenscheibentaster gedrückt wird und kein neues Startsignal vorhanden ist.

ZEIT-DIAGRAMM: NOCKENSCHLEIBENSCHRITTMOTOR



2.3.2.2. Richtungssteuerung der Motoren

Die Drehrichtung der Motoren wird über je ein Relais gesteuert. An die Relais (je zweimal UM), die als Umpoler geschaltet sind, sind auch noch Endlagentaster für den X- und Y-Schlitten angeschlossen. Diese Endlagentaster werden auch für das Initialisieren des Plotters benötigt.

2.3.2.3. Schreibstiftsteuerung

Der Magnet für den Stift (PEN) wird von einem 5. Ausgang des Interfaces über einen Verstärker geschaltet.

2.3.2.4. Signalhupensteuerung

Da der Plotter für kompliziertere Bilder doch sehr lange Zeit benötigt (bis zu mehreren Stunden), habe ich eine kleine Hupe direkt an einen Ausgang des Interfaces angeschlossen. Diese Hupe wird vom Computer dann in Betrieb gesetzt, wenn die betreffende Zeichnung fertig ist.

3. SOFTWARE

3.1. Aufgabe

Das BASIC-Demonstrationsprogramm soll den Plotter ansteuern und mit ihm Geraden, Kreise und andere Figuren zeichnen. Die Ansteuerung des Plotters soll innerhalb eines Rasters von 200 x 290 Punkten erfolgen. Dafür benötigt man bei meinem Plotter zwei verschiedene Programmteile. Als erstes braucht man ein Programm, das die Figuren berechnet. Dieses muß für jeden Kurvenpunkt geschehen. Zweitens braucht man ein Programm, das die Motoren, die Relais und den Stift (PEN) ansteuert.

3.2. Probleme

Nennenswerte Probleme gab es eigentlich nur bei den Unterprogrammen für die Linearisierung und für die Ansteuerung der Motoren, Relais und des Stiftes (Ausgabeprogramm). Ein Problem bei dem Ausgabeprogramm war ein kleiner logischer Fehler. Ich hatte vergessen, eine Variable auf 0 zu setzen (siehe Zeile 5073 - 5090). Dadurch wurden die Motoren oft versehentlich gestartet. Weiterhin stellte sich nach einiger Zeit heraus, daß die Dauer des Startimpulses, die über eine FOR TO NEXT - Schleife erzeugt wurde, sehr stark schwankte (100 ms bis 200 ms). Dieses rührte vom BASIC-Interpreter her. Um dieses Problem zu lösen, rufe ich jetzt ein Assembler-Unterprogramm auf. Es hat eine Verzögerungszeit von 150 ms.

Im Linearisierungsprogramm waren mehrere kleine logische Fehler vorhanden. Ich hatte erstens vergessen, die Stufenlänge in der Y-Richtung mit dem richtigen Vorzeichen zu versehen, denn dieses Vorzeichen ist von der Bewegungsrichtung in der X-Achse abhängig. Zweitens hatte ich vergessen, die Linearisierungslogik zu überspringen, wenn in der X-Richtung keine Bewegung ausgeführt werden soll (/0 Fehler).

3.3. Beschreibung der Software

3.3.1. Programmübersicht

Das Programm ist in mehrere Unterprogramme aufgeteilt. Dies sind einmal die Programme, die den Plotter direkt ansteuern: Das Initialisierungsprogramm, das Linearisierungsprogramm, das Ausgabeprogramm und das Piepston-Programm. - Zum anderen gibt es noch die Programme, die die Kurvenpunkte berechnen. Dieses sind:

1. Tabellenprogramm
2. Kreise, Ellipsen und Lissayousfiguren
3. Testprogramm
4. Zufallsprogramm
5. Funktionen
6. eigenes Programm

Diese Programme werden von einem Hauptprogramm aufgerufen, das auch die Initialisierung aufruft.

3.3.2. Beschreibung der einzelnen Programme

3.3.2.1. Hauptprogramm (Zeile 5 - 220)

Das Hauptprogramm fragt ab, ob initialisiert werden soll und führt dieses dann bei Bedarf aus. Danach fragt es ab, ob die Auswahl aller möglichen Programme bekannt ist, zeigt diese Auswahl bei Bedarf an und springt dann in das gewünschte Programm. Nach der Ausführung kehrt es an den Anfang des Hauptprogramms zurück.

3.3.2.2. Programme zur Berechnung der Kurvenpunkte

3.3.2.2.1. Tabellenprogramm (Zeile 6500 - 6790/Zeile 2000 - 2410)

Dieses Programm legt eine Tabelle an, in die einzelne Kurvenpunkte eingespeichert werden können. Die Eingabe der Koordinaten erfolgt entweder als Zahlenwert über die Tastatur, oder die Punkte werden mit einem Knüppel angefahren. Während der Ausgabe werden die einzelnen Koordinaten aufgelistet.

3.3.2.2.2. Kreise, Ellipsen und Lissayous-Figuren (Zeile 7000-7440)

Dieses Programm ist in zwei Hauptteile aufgeteilt.

Zuerst kommt ein Programm, in dem die Abmessungen der Figuren angegeben werden. Wenn ein Kreis gezeichnet werden soll, werden die Koordinaten des Mittelpunktes und der Radius gefragt. Bei einer Ellipse wird nach dem Mittelpunkt und den Halbachsen gefragt. Dieses ermöglicht jedoch nur waagerechte und senkrechte Ellipsen. Bei den Lissayous-Figuren wird der Mittelpunkt, die Höhe und die Breite, der Frequenzfaktor X/Y und die Phasenverschiebung von X zu Y in Grad gefragt.

Danach folgt das Programm zum Ausrechnen der einzelnen Kurvenpunkte (dieses Programm kann auch von einem anderen Programm aufgerufen werden). Es braucht folgende Variablen:

XM, YM: Mittelpunkt der Figur

XF, YF: Frequenzfaktor

XE, YE: Größe

VY: Phasenverschiebung von X zu Y (Bogenmaß)

Es rechnet damit dann die einzelnen Kurvenpunkte aus und läßt diese mit Linien verbinden.

Wird ein Punkt errechnet, dessen Koordinaten den Koordinaten des Anfangspunktes ähneln (bis zu zwei Schritten in jede Richtung abweichend), fragt dieses Programm ab, ob es noch weiterzeichnen soll.

3.3.2.2.3. Testprogramm (Zeile 1000 - 1070)

Im Testprogramm werden nur einzelne Koordinaten und der Zustand des Stiftes eingegeben, und diese dann durch Linien verbunden.

3.3.2.2.4. Zufallsprogramm (Zeile 1500 - 1640)

Das Zufallsprogramm erwartet als Eingabe die Koordinaten der linken unteren Ecke, die Breite und die Höhe sowie die Anzahl der Linien, die gezeichnet werden sollen. Dann denkt es sich mit einem Zufallsgenerator die Koordinaten von Punkten in diesem Bereich aus und verbindet sie. Während des Programmes werden die Koordinaten der einzelnen Punkte aufgelistet.

3.3.2.2.5. Funktionen (Zeile 7500 - 7720)

Das Funktionsprogramm dient zum Aufzeichnen einzelner Funktionen mit dem Plotter. In ihm müssen die Programmzeilen, in denen die X und Y-Koordinaten ausgerechnet werden, selbst geschrieben werden (Zeile 7650 und 7660). Es stehen dabei folgende Variablen zur Verfügung, die beim Beginnen dieses Programmes eingegeben werden müssen:

XP, YP (Nullpunkt)

XG, YG (Größe)

XF, YF (Faktor)

In den selbstgeschriebenen Programmzeilen müssen dann die Werte für XL und YL ausgerechnet werden. Es steht ein Zähler Q, der von 0 bis 1 zählt, zur Verfügung.

3.3.2.2.6. Eigenes Programm (Zeile 8000 - 9000 möglich)

Dieses eigene Programm kann vollständig selbst geschrieben werden und dient dazu, nicht häufig vorkommende Kurven zu zeichnen, die man mit dem Funktionsprogramm nicht mehr vernünftig eingeben kann. Dieses Programm muß mit "Return" abgeschlossen werden.

3.3.2.3. Ansteuerungsprogramme für den Plotter

3.3.2.3.1. Initialisierungsprogramm (Zeile 5500 - 5570)

Das Initialisierungsprogramm läßt den Stift auf den Null-Punkt zurückfahren. Dort hält er durch die Endlagentaster an. Antwortet man auf die Frage "Stift links unten?" mit "J" wird dieser Punkt als der Null-Punkt festgesetzt.

3.3.2.3.2. Linearisierungsprogramm (Zeile 6000 - 6150)

Dieses Programm hat die Eingangsvariablen XL und YL. Mit diesen Variablen wird die Y-Schrittweite ausgerechnet. Der X-Wert hat die Schrittweite 1. Ist der jeweils neue Punkt der Geraden errechnet worden, so wird mit diesen Koordinaten das Ausgabeprogramm aufgerufen. Wenn in der X-Achse keine Bewegung vorhanden ist, wird direkt in das Ausgabeprogramm gesprungen. Zwei Schritte vor Erreichen des Endpunktes wird dieser Punkt direkt angefahren.

3.3.2.3.3. Ausgabeprogramm (Zeile 5000 - 5390)

Das Ausgabeprogramm hat als Eingangsvariable XN, YN und P. Es testet zuerst, ob sich die Koordinaten in dem möglichen zu zeichnenden Bereich befinden. Ist dieses nicht der Fall, wird ein Meldetext ausgegeben und zurück ins Hauptprogramm gesprungen. Danach wird abgetestet, ob sich der Wert der Variablen P zwischen 1 und 0 jeweils einschließlich befindet. Ist dies nicht der Fall, wird ebenfalls ein Meldetext ausgegeben und zurück ins Hauptprogramm gesprungen. Nun werden alle Werte gerundet (ganzzahliger Anteil). Es wird jetzt der Zustand des PENS auf das Tor ausgegeben. Nun wird die neue Richtung bestimmt und auf die Relais ausgegeben. Danach wird errechnet, ob die Motoren gestartet werden sollen und das Startsignal für eine bestimmte Dauer ausgegeben. Nun wird das Signal wieder zurückgenommen, und es wird erneut eine kurze Zeit gewartet (siehe Skizze Seite 4). Dann wird die neue Position errechnet (XA, YA). Wenn jetzt der Zielpunkt noch nicht erreicht ist, wird das Programm nochmals durchlaufen. Bei Erreichen des Zielpunktes endet es (Zeile 5210).

3.3.2.3.4. Piepston-Programm (Zeile 5800 - 5900)

Dieses Programm gibt für eine bestimmte Dauer ein Signal auf den Ausgang. Der Ton wird im Sekundenabstand solange wiederholt, bis eine Taste gedrückt wird.

4. L I T E R A T U R U N D H I L F E N

4.1. benutzte Literatur:

1. Beschreibung des Plotters WX/4671 von Watenabe GmbH
2. Aufsatz in der Zeitschrift "CHIP" No. 5, Mai 1981
Seite 37/38 von Peter Prisille
Titel: "Plotter selbst gebaut"
3. Aufsatz in der Zeitschrift "CHIP" No. 6, Juni 1981
Seite 78/79, Titel: "Ein intelligenter Zeichenstift"

4.2. Hilfen


beratend: mein Vater Heinz Purnhagen
Netzteile: Erstattung der Kosten durch den
 Senator für Bildung, Bremen

5. A N H A N G

bestehend aus:



- | | |
|-------------------------------------|-------------|
| 5.1. Schaltzeichnung | A 1 - A 2 |
| 5.2. Fotos vom Plotter | A 2 - A 4 |
| 5.3. Programm | A 5 - A 10 |
| 5.4. vom Plotter gezeichnete Bilder | A 11 - A 20 |


SCHALTUNG: PLOTTER




H = HUPE ET = ENDLAGENTASTER

SCHALTUNG: KNÜPPEL








ERSTE
VERSION
MEINES
PLOTTERS




DRAUFSICHT
AUF DEN
FERTIGEN
PLOTTER



SCHREIBSTIFT →



↑
Y-MOTOR →



PROGRAMME

ZU 3.3.2.1. HAUPTPROGRAMM

```
5 OUT 133,0
10 PRINT CHR$(12)
20 PRINT " P L O T T E R "
25 PRINT
30 PRINT "          25.1.82"
35 PRINT
40 PRINT "ZULAESSIGE GROESSEN:"
50 PRINT "X=0-290 Y=0-200 P=0-1"
55 PRINT
60 PRINT "BITTE ANTWORTEN MIT:"
70 PRINT " JA=J  NEIN=N"
75 PRINT
80 INPUT "INITIALISIEREN";G$
90 IF G$="J" THEN GOSUB 5500
100 INPUT "KENNEN SIE DIE AUSWAHL";G$
110 IF G$="J" THEN 200
120 PRINT CHR$(12)
130 PRINT "  AUSWAHL "
140 PRINT
150 PRINT "TABELLENPROGRAMM =1"
160 PRINT "KREISE, ELLIPSEN UND LISSAYOUSFIGUREN =2"
170 PRINT "TESTPROGRAMM =3"
180 PRINT "ZUFALLSPROGRAMM =4"
190 PRINT "FUNKTIONEN =5"
195 PRINT "EIGENES PROGRAMM =6"
200 INPUT "WELCHES PROGRAMM (CODE)";G
205 PRINT CHR$(12)
210 ON G GOSUB 6500,7200,1000,1500,7500,8000
220 GOTO 80
```

ZU 3.3.2.2.1. TABELLENPROGRAMM

```
6500 REM TABELLE
6510 PRINT "TABELLENPROGRAMM"
6520 INPUT "SPEICHERINHALT PLOTTEN";G$
6530 IF G$="J" THEN 6700
6533 INPUT "EINGEBEN";G$
6537 IF G$="N" THEN 6770
6540 PRINT "BITTE EINGEBEN"
6542 INPUT "KNUEPPEL";G$
6544 IF G$="J" THEN 2000
6546 INPUT "MIT AUSGABE";MA$
6550 INPUT "WIEVIELE PUNKTE";M
6551 LET M=M-1
6552 PRINT "ENDE=999 IN XL"
6555 POKE 33536,M
6560 FOR D=33537 TO 33537+M*3 STEP 3
6565 PRINT "BITTE NR. ";((D-33537)/3+1)
6570 INPUT "XL";XL
6575 IF XL=999 THEN POKE 33536,(D-33537)/3-1 : GOTO 6500
6580 INPUT "YL";YL
6590 INPUT "P";P
6593 IF MA$="N" THEN 6600
6597 GOSUB 6000
6598 OUT 133,0
6600 INPUT "OK";G$
6610 IF G$="N" THEN 6565
```

```

6620 POKE D,XL=INT(XL/256)*256
6630 POKE D+1,INT(XL/256)+P*2
6640 POKE D+2,YL
6645 OUT 133,0
6650 NEXT D
6660 GOTO 6500

```

TABELLENAUSGABE

```

6700 LET M=PEEK(33536)
6702 PRINT M+1;"PUNKTE"
6704 FOR I=1 TO 2000
6705 NEXT I
6706 GET G$
6707 IF G$<>"N" THEN 6710
6708 INPUT "WIEVIEL PUNKTE";M
6709 LET M=M-1 : POKE 33536,M
6710 FOR D=33537 TO 33537+M*3 STEP 3
6720 LET XL=PEEK(D)+(PEEK(D+1) AND 1)*256
6730 LET YL=PEEK(D+2)
6740 LET P=(PEEK(D+1) AND 2)/2
6745 PRINT "XL";XL;"YL";YL;"P";P
6750 GOSUB 6000
6760 NEXT D
6765 OUT 133,0
6770 PRINT "NOCH ETWAS?"
6775 GOSUB 5800
6780 IF G$="J" THEN 6500
6790 RETURN

```

KNÜPPEL

```

2000 REM KNUEPPEL
2005 LET P=0
2010 LET XL=XA
2020 LET YL=YA
2030 LET D=33537
2040 LET IN=INP(132)
2045 GET G$
2050 IF G$="E" THEN 2300
2060 IF D>33786 THEN LET D=33786
2065 IF D<33537 THEN LET D=33537
2070 IF (IN AND 32)=0 THEN 2200
2080 IF G$="V" THEN LET D=D+3
2090 IF G$="R" THEN LET D=D-3
2100 LET P=1-((IN AND 16)/16)
2110 IF (IN AND 1)=0 THEN LET YL=YL+1
2120 IF (IN AND 2)=0 THEN LET XL=XL+1
2130 IF (IN AND 4)=0 THEN LET YL=YL-1
2140 IF (IN AND 8)=0 THEN LET XL=XL-1
2150 IF XL>290 THEN LET XL=290
2160 IF XL<0 THEN LET XL=0
2170 IF YL>200 THEN LET YL=200
2180 IF YL<0 THEN LET YL=0
2185 GOSUB 2350
2190 GOSUB 6000
2195 GOTO 2040
2200 POKE D,XL=INT(XL/256)*256
2210 POKE D+1,INT(XL/256)+P*2
2220 POKE D+2,YL
2230 LET IN=INP(132)
2240 IF (IN AND 32)=0 THEN 2230
2250 LET D=D+3
2260 GOTO 2100

```

Hohe Schule für Technik Bremen

```

2300 POKE 33536,(D-33537)/3-1
2305 PRINT CHR$(12)
2310 GOTO 6500
2350 LET XM=PEEK(D)+(PEEK(D+1) AND I)*256
2360 LET YM=PEEK(D+2)
2370 LET PM=(PEEK(D+1) AND 2)/2
2380 LET NR=(D-33537)/3+1
2385 PRINT CHR$(12)
2390 PRINT "XL";XL;"YL";YL;"P";P
2400 PRINT "NR";NR;"XM";XM;"YM";YM;"PM";PM
2410 RETURN

```

ZU 3.3.2.2.2. KREIS, ELLIPSEN UND LISSAYOUSFIGUREN (EINGABE)

Ree

```

7200 PRINT "KREISE, ELLIPSEN UND LISSAYOUSFIGUREN"
7210 INPUT "K=1, E=2, L=3";G
7220 ON G GOTO 7250,7310,7360
7230 PRINT "DIESE EINGABE IST FALSCH"
7240 GOTO 7200
7250 PRINT "KREIS"
7260 INPUT "MITTELPUNKT(X,Y)";XM,YM
7270 INPUT "RADIUS";R
7280 LET XE=R; LET YE=R
7290 LET VY=1.57073; LET XF=1; LET YF=1
7300 GOTO 7410
7310 PRINT "ELLIPSE"
7320 INPUT "MITTELPUNKT(X,Y)";XM,YM
7330 INPUT "HALBACHSEN(X,Y)";XE,YE
7340 LET VY=1.57073; LET XF=1; LET YF=1
7350 GOTO 7410
7360 PRINT "LISSAYOUSFIGUREN"
7370 INPUT "MITTELPUNKT(X,Y)";XM,YM
7380 INPUT "BREITE, HOEHE(X,Y)";XE,YE
7385 LET XE=XE/2; LET YE=YE/2
7390 INPUT "FREQUENZ-FAKTOR(X,Y)";XF,YF
7400 INPUT "PHASENVERSCHIEBUNG Y(GRAD)";VY
7405 LET VY=VY/57.2958
7410 GOSUB 7000
7420 INPUT "NOCH ETWAS";G$
7430 IF G$="J" THEN 7200
7440 RETURN

```

(AUSGABE)

u m Hochschule für Technik Bremen

```

7000 REM KREISE, ELLIPSEN UND LISSAYOUSFIGUREN
7010 REM VARIABLEN: XM, YM, XF, YF, XE, YE, VY
7020 LET V=1/((XE+YE)/2)
7030 LET U=0-V
7040 LET U=U+V
7050 LET XL=INT((SIN(U*XF)*XE)+XM)
7060 LET YL=INT((SIN(U*YF+VY)*YE)+YM)
7070 LET P=1
7080 IF U=0 THEN LET P=0; LET XR=XL; LET YR=YL
7090 GOSUB 6000
7095 IF U<V*5 THEN 7160
7100 FOR I=-2 TO +2
7110 IF XL+I=XR THEN 7130
7120 NEXT I
7125 GOTO 7040
7130 FOR I=-2 TO +2
7140 IF YL+I=YR THEN 7170
7150 NEXT I

```

7160 GOTO 7040
7170 OUT 133,0
7180 PRINT "WEITER?"
7185 GOSUB 5800
7190 IF G#="J" THEN 7040
7195 RETURN

ZU 3.3.2.2.3 TESTPROGRAMM

1000 PRINT "TESTPROGRAMM"
1010 INPUT "XL";XL
1020 INPUT "YL";YL
1030 INPUT "P";P
1040 GOSUB 6000
1045 OUT 133,0
1050 PRINT "WEITER?"
1055 GOSUB 5800
1060 IF G#="J" THEN 1010
1070 RETURN

ZU 3.3.2.2.4 ZUFALLSPROGRAMM

1500 PRINT "ZUFALLSPROGRAMM"
1510 INPUT "LINKE UNTERE ECKE(X,Y)";XE,YE
1520 INPUT "BREITE,HOEHE(X,Y)";XG,YG
1530 INPUT "WIEVIELE LINIEN";G
1540 LET XL=XE : LET YL=YE : LET P=0
1550 GOSUB 6000
1560 FOR II=1 TO G
1570 LET XL=INT(RND(1)*XG+XE)
1580 LET YL=INT(RND(1)*YG+YE)
1590 LET P=P+1
1595 PRINT "XL";XL;"YL";YL
1600 GOSUB 6000
1610 NEXT II
1615 OUT 133,0
1620 PRINT "WEITER?"
1625 GOSUB 5800
1630 IF G#="J" THEN 1500
1640 RETURN

ZU 3.3.4.4.5 FUNKTIONEN

7500 PRINT "FUNKTIONEN"
7510 PRINT "7650,7660,XP,YF,XG,YG,XF,YF"
7600 INPUT "NULLPUNKT(X,Y)";XP,YF
7610 INPUT "GROESSE(X,Y)";XG,YG
7620 INPUT "FAKTOR(X,Y)";XF,YF
7630 FOR Q=0 TO 1 STEP 0.005
7640 LET P=P+1
7650 LET XL=Q*XG+XP
7660 LET YL=(SIN(Q*6.28318*XF)+SIN(Q*6.28318*YF))*YG+YF
7663 IF YL<0 THEN LET YL=0 : LET P=P+1
7667 IF YL>200 THEN LET YL=200 : LET P=P+1
7670 IF Q=0 THEN LET P=P+1
7680 GOSUB 6000
7690 NEXT Q
7695 OUT 133,0
7700 PRINT "NOCH ETWAS?"
7705 GOSUB 5800
7710 IF G#="J" THEN 7500
7720 RETURN

ZU 3.3.2.2.6 EIGENES PROGRAMM

8000 PRINT "EIGENES PROGRAMM"
8010 PRINT "SINUS-MANTEL"
8020 INPUT "GROESSE(X,Y)"; XG, YG
8030 INPUT "NULLPUNKT(X,Y)"; XP, YP
8040 INPUT "VON, BIS, SCHRITTW. (2-4)"; VO, BI, SC
8050 FOR SZ=VO TO BI STEP SC
8060 FOR ZZ=0 TO 1 STEP 0.005
8070 LET XL=ZZ*XG+XP
8080 LET YL=SIN(ZZ*6.28318*SZ)*(1/2*(ZZ*5))*((SZ-2)/2)*YG+YP
8090 LET P=1
8100 IF ZZ=0 THEN LET P=0
8110 GOSUB 6000
8120 NEXT ZZ
8130 NEXT SZ
8140 PRINT "NOCH ETWAS?"
8150 GOSUB 5800
8160 IF G\$="J" THEN GOTO 8000
8170 RETURN

Technik für Technik Bremen

ZU 3.3.2.3.1 INITIALISIERUNG

5500 REM INITIALISIERUNG
5505 OUT 133,27
5510 OUT 133,27
5515 INPUT "STIFT LINKS UNTEN"; H\$
5520 IF H\$="J" THEN 5540
5530 GOTO 5505
5540 OUT 133,0
5550 LET XA=0
5560 LET YA=0
5570 RETURN

ZU 3.3.2.3.2. LINEARISIERUNG

6000 REM LINIEN
6010 IF ABS(XL-XA)<4 THEN 6120
6020 IF ABS(YL-YA)<4 THEN 6120
6030 IF XL=XA THEN 6120
6040 LET B=(YL-YA)/(XL-XA)*SGN(XL-XA)
6050 LET YB=YA
6060 LET Z=SGN(XL-XA)
6070 FOR XN=XA+Z TO XL-Z-Z STEP Z
6080 LET YB=YB+B
6090 LET YN=INT(YB+.5)
6100 GOSUB 5000
6110 NEXT XN
6120 LET XN=XL
6130 LET YN=YL
6140 GOSUB 5000
6150 RETURN

Technik für Technik E

ZU 3.3.2.3.3 AUSGABE

Rechenzentrum Hochsch.

```
5000 REM AUSGABE
5001 IF XN<0 THEN 5300
5002 IF XN>290 THEN 5300
5004 IF YN<0 THEN 5300
5005 IF YN>200 THEN 5300
5006 IF (INT(P/2))<>0 THEN 5350
5007 LET XN=INT(XN)
5008 LET YN=INT(YN)
5009 LET P=INT(P)
5010 LET X=XA-XN
5020 LET Y=YA-YN
5025 LET O=(O AND 9)+P*64
5030 OUT 133,0
5040 LET XR=INT(SGN(X)*0.5+0.5)
5045 IF X=0 THEN LET XR=(O AND 1)
5050 LET YR=INT(SGN(Y)*0.5+0.5)
5055 IF Y=0 THEN LET YR=(O AND 8)/8
5060 LET O=XR+YR*8+P*64
5070 OUT 133,0
5073 LET XS=0
5077 LET YS=0
5080 IF X<>0 THEN LET XS=1
5090 IF Y<>0 THEN LET YS=1
5100 LET O1=O+XS*2+YS*16
5110 OUT 133,O1
5120 LET HL=USR(3582)
5140 OUT 133,0
5150 LET HL=USR(3582)
5170 LET XA=XA-SGN(X)
5180 LET YA=YA-SGN(Y)
5190 IF XA<>XN THEN 5010
5200 IF YA<>YN THEN 5010
5210 RETURN
```

} 150ms VERZÖGERUNG


MELDETEXTE

```
5300 PRINT "WAS WOLLEN SIE BEIM EIFFELTURM?"
5310 PRINT "DIESES KOORDINATENPAAR"
5320 PRINT "IST NICHT ZULAESSIG!"
5325 PRINT "XN";XN;"YN";YN
5327 OUT 133,0
5330 GOTO 80
5350 PRINT "DIESEN ZUSTAND DES PENS"
5360 PRINT "GIBT ES NICHT"
5370 PRINT "P";P
5380 OUT 133,0
5390 GOTO 80
```


ZU 3.3.2.3.4. PIEPSTON

```
5800 REM PIEPSTON
5820 OUT 133,128
5830 LET HL=USR(3582)
5840 LET HL=USR(3582)
5850 OUT 133,0
5860 GET G$
5870 IF G$="" THEN GOTO 5880
5875 RETURN
5880 LET HL=USR(3582)
5890 LET HL=USR(3582)
5900 GOTO 5820
```


20. 10. 1




ZUCCHETTI S. R. L. TORINO





27.7.84






△ A2








A-16




SCHWÜBUNG

10.17.21


Handwritten text in a vertical column, possibly a list or index, consisting of approximately 20 entries.




6.12.81



$y = f(x)$



79.9.5-1
ZUFALLSPROGRAMM



7.9.81
TEST PLOTS

A20